

# Tips and Tricks to Create In-text Tables in Clinical Trial Reporting Using SAS



 Shafi Consultancy Ltd.

# Overview

- In-text tables in CTR
- Production of in-text tables for CTR
- SAS built-in Templates
- PROC TEMPLATE code
- Example 1: Summary of Baseline Characteristics
- Example 2: Summary of Baseline Characteristics with a changed template

# In-text tables in CTR

- Produced by Medical Writers in the past
- Increasingly the task is being handed over to the programmers
- Generally pre-planned and produced at the same time as when the other TLFs are produced for the CTR appendix.
  
- Aim:
  - Show how to create an in-text table using SAS ODS RTF destination with PROC REPORT and PROC TEMPLATE

# Production of in-text tables for CTR

- There are many different styles of templates in SAS catalog
- Use an existing template to generate a new one by changing different attributes, e.g.
  - Title and footnote fonts, Body fonts
  - Attributes for row headings and column headings
  - Background color, Link color
  - Internal and external border color and thickness
  - Page settings for display result

# SAS Built-in Templates

To see all the SAS built-in style templates anyone can use the following code segment:

```
PROC TEMPLATE;  
  LIST styles;  
RUN;
```

The SAS System

Listing of: SASHELP.TMPLMST

Path Filter is: Styles

Sort by: PATH/ASCENDING

Obs	Path	Type
1	Styles	Dir
2	Styles.Analysis	Style
3	Styles.BarrettsBlue	Style
4	Styles.BlockPrint	Style
5	Styles.DTree	Style
6	Styles.Daisy	Style
7	Styles.Default	Style
8	Styles.Dove	Style
9	Styles.EGDefault	Style
10	Styles.Excel	Style
11	Styles.FancyPrinter	Style
12	Styles.Festival	Style

# PROC TEMPLATE Code

## Header, Title and Footnote of Word table

```
PROC TEMPLATE;
```

```
  DEFINE STYLE Styles.RTF1; /* Defining Template Name */
```

```
  PARENT = Styles.RTF;      /* Parent Style Template */
```

```
  REPLACE fonts /
```

```
    /* Titles from TITLE statements */
```

```
    'TitleFont' = ("Calibri, Arial, Times Roman", 12pt)
```

```
    /* Table column and row headings */
```

```
    'headingFont' = ("Arial, Times Roman", 10pt, Bold)
```

```
    /* Data in table cells */
```

```
    'docFont' = ("Arial, Times Roman", 10pt)
```

```
    /* Footnotes from FOOTNOTE statements */
```

```
    'footFont' = ("Calibri, Times Roman", 10pt);
```

# PROC TEMPLATE Code

## Header, Title and Footnote of Word table

```
/* Links, text color, row/column header, page background color */
REPLACE color_list /
    'link' = blue
    'fg'    = black
    'bgH'   = white
    'bg'    = white;

/* Header Attributes */
CLASS Header /    verticalalign = top;

/* Title and Footnote Attributes */
CLASS TitlesAndFooters /
    font        = Fonts('TitleFont')
    color       = color_list('link')
    textalign   = left
    frame       = void;
```

# PROC TEMPLATE Code

## Main body of Word table

```
/* Page setting for display results */
REPLACE Body from Document /
    bottommargin = .7in          topmargin = .7in
    rightmargin  = .7in          leftmargin = .7in;

/* internal and outside borders, the width of the borders and rules */
REPLACE Table from Output /
    frame      = box
    rules      = all
    borderwidth = 1pt
    bordercolor = color_list('fg')
    width      = 100%  asis = on;

END;
RUN;
```



# Example 1:

## Summary of Baseline Characteristics

### ODS RTF and PROC REPORT

```
/* Define page orientation, option to not print date and page number */
/* ODS destination is closed for the Listing */
/* An escape character is specified to perform in-line formatting */
OPTIONS NODATE NONUMBER ORIENTATION=PORTRAIT;
ODS LISTING CLOSE;
ODS ESCAPECHAR = '^';

/* Specify destination path for RTF file and select a style template */
/* Define device type for RTF (only supports: SASSEMFM, PNG and JPEG) */
/* Define titles, footnotes and page numbers (i.e. page x of y)*/
ODS RTF FILE="&pathname./t_demo.rtf" STYLE=styles.RTF1 DEVICE=SASSEMFM;
TITLE 'Table: Patients baseline characteristics';
FOOTNOTE J=CENTER 'Page ^{thispage} of ^{lastpage}';
```

# Example 1:

## Summary of Baseline Characteristics

### ODS RTF and PROC REPORT

```
PROC REPORT DATA=final HEADLINE LS=140 PS=50 SPACING=3 SPLIT='$';
  COLUMN (decodeord block decode col1 col2 coltot);

  DEFINE decodeord / ORDER ORDER=INTERNAL NOPRINT;
  DEFINE block      / ORDER ORDER=INTERNAL NOPRINT;
  DEFINE decode     / ' ' FLOW STYLE={WIDTH=27%};
  DEFINE col1      / "DummyRed$(N =&col1.)" WIDTH=26 STYLE={WIDTH=12%};
  DEFINE col2      / "DummyGreen$(N =&col2.)" WIDTH=26 STYLE={WIDTH=12%};
  DEFINE coltot    / "Total$(N =&coltotal.)" WIDTH=26 STYLE={WIDTH=10%};
  /*Include conditional format for specific rows, columns, footnotes */
  .....
RUN;
ODS RTF CLOSE;
ODS LISTING;
OPTIONS DATE NUMBER;
```

# Example 1:

## Output produced in the RTF format

Table: Patients baseline characteristics

	DummyRed (N =136)	DummyGreen (N =166)	Total (N =302)
<b>Age (years), mean (SD)</b>	42.7 (12.67)	44.0 (13.25)	43.4 (12.99)
<b>Weight (kg)</b>			
Mean (SD)]	85.8 (20.30)	89.4 (24.75)	87.8 (22.90)
Median (Min, Max)	83.5 ( 47, 149)	83.0 ( 48, 200)	83.5 ( 47, 200)
<b>Weight category</b>			
<= 100 kg	104 (77.0)	121 (72.9)	225 (74.8)
> 100 kg	31 (23.0)	45 (27.1)	76 (25.2)
<= 90 kg	83 (61.5)	103 (62.0)	186 (61.8)
> 90 kg	52 (38.5)	63 (38.0)	115 (38.2)
<b>Gender</b>			
Female	46 (33.8)	54 (32.5)	100 (33.1)
Male	90 (66.2)	112 (67.5)	202 (66.9)
<b>BMI, mean (SD)</b>	28.8 (5.55)	29.7 (6.97)	29.3 (6.38)
<b>Country</b>			
Europe (Total)	111 (81.6)	139 (83.7)	250 (82.8)
Western EU	83 (61.0)	106 (63.9)	189 (62.6)
Eastern EU	28 (20.6)	33 (19.9)	61 (20.2)
Non-EU	25 (18.4)	27 (16.3)	52 (17.2)

Sources: ADSL;

Page 1 of 1

# Example 2:

## Summary of Baseline Characteristics with changed template

In the PROC TEMPLATE we will update the following code in red -

```
/* Header Attributes */  
CLASS Header /  
    foreground = White;  
  
/* Row and Column header background */  
REPLACE color_list /  
    'bgH' = cx4F81BD  
  
/* Borders color*/  
REPLACE Table from Output /  
    bordercolor = color_list('bgH')
```

# Example 2:

## Output produced with the new changes

Table: Patients baseline characteristics

	DummyRed (N =136)	DummyGreen (N =166)	Total (N =302)
<b>Age, mean (SD)</b>	42.7 (12.67)	44.0 (13.25)	43.4 (12.99)
<b>Weight</b>			
Mean (SD)	85.8 (20.30)	89.4 (24.75)	87.8 (22.90)
Median (Min, Max)	83.5 ( 47, 149)	83.0 ( 48, 200)	83.5 ( 47, 200)
<b>Weight category</b>			
<= 100 kg	104 (77.0)	121 (72.9)	225 (74.8)
> 100 kg	31 (23.0)	45 (27.1)	76 (25.2)
<= 90 kg	83 (61.5)	103 (62.0)	186 (61.8)
> 90 kg	52 (38.5)	63 (38.0)	115 (38.2)
<b>Gender</b>			
Female	46 (33.8)	54 (32.5)	100 (33.1)
Male	90 (66.2)	112 (67.5)	202 (66.9)
<b>BMI, mean (SD)</b>	28.8 (5.55)	29.7 (6.97)	29.3 (6.38)
<b>Country</b>			
Europe (Total)	111 (81.6)	139 (83.7)	250 (82.8)
Western EU	83 (61.0)	106 (63.9)	189 (62.6)
Eastern EU	28 (20.6)	33 (19.9)	61 (20.2)
non-EU	25 (18.4)	27 (16.3)	52 (17.2)

Sources: ADSL;

# Acknowledgement

Thank you to Md. Jabrul Islam for all his support in preparing this paper and presentation.

# Questions or Comments?



Please feel free to contact

**Rafi Rahi**

[rafi@shaficonsultancy.com](mailto:rafi@shaficonsultancy.com)

[www.shaficonsultancy.com](http://www.shaficonsultancy.com)

 Shafi Consultancy Ltd.